

Tensor perturbations of Einstein's equations

Preliminaries

Load tensor package

```
In[2]:= << xAct`xPand`
```

```
-----
Package xAct`xPerm` version 1.2.3, {2015, 8, 23}
Copyright (C) 2003–2020, Jose M. Martin-Garcia, under the General Public License .
Connecting to external linux executable ...
Connection established .
-----
```

```
-----
Package xAct`xTensor` version 1.1.4, {2020, 2, 16}
Copyright (C) 2002–2020, Jose M. Martin-Garcia, under the General Public License .
-----
```

```
-----
Package xAct`xPert` version 1.0.6, {2018, 2, 28}
Copyright (C) 2005–2018, David Brizuela, Jose M. Martin-Garcia
and Guillermo A. Mena Marugan, under the General Public License .
** Variable $PrePrint assigned value ScreenDollarIndices
** Variable $CovDFormat changed from Prefix to Postfix
** Option AllowUpperDerivatives of ContractMetric changed from False to True
** Option MetricOn of MakeRule changed from None to All
** Option ContractMetrics of MakeRule changed from False to True
-----
```

```
-----
Package xAct`xPand` version 0.4.3, {2019, 3, 4}
Copyright (C) 2012–2018, Cyril Pitrou,
Xavier Roy and Obinna Umeh under the General Public License .
-----
```

```
-----
These packages come with ABSOLUTELY NO WARRANTY ; for details type
Disclaimer []. This is free software, and you are welcome to redistribute
it under certain conditions . See the General Public License for details .
-----
```

Nicer printing

```
In[3]:= $PrePrint = ScreenDollarIndices ;
In[4]:= $CovDFormat = "Prefix";
In[5]:= $DefInfoQ = False ;
```

Object definitions

Spacetime manifold

The spacetime manifold M , on which tensors will be defined. Some Greek letters are defined as tangent space indices.

```
In[7]:= DefManifold [M, 4, { $\alpha$ ,  $\beta$ ,  $\gamma$ ,  $\zeta$ ,  $\lambda$ ,  $\mu$ ,  $\nu$ ,  $\omega$ ,  $\tau$ ,  $\sigma$ }
```

Metric

The background metric g of signature $(-,+,+,+)$. The Levi-Civita derivative of a tensor A_μ will be written as $\nabla_\mu A_\nu$ in prefix notation or $A_{\nu;\mu}$ in postfix notation. Note that this is not the physical metric; the latter will be defined later as a^2 times this metric, where a is the scale factor.

```
In[8]:= DefMetric [-1, Met[- $\alpha$ , - $\beta$ ], CD, {";", "∇"}, PrintAs → "g"]
```

FLRW background geometry

We now define the cosmologically symmetric background geometry. Here we choose a spatially curved Friedmann-Lemaître-Robertson-Walker background metric. The physical metric, which will be called `gah2`, is defined automatically by `xPand`.

```
In[9]:= SetSlicing [Met, Orth, SMet, SD, {"I", "D"}, "FLCurved"]
Rules {1, 2, 3, 4, 5, 6, 7, 8} have been declared as UpValues for SMet.
Rules {1, 2, 3, 4, 5, 6, 7, 8} have been declared as UpValues for SMet.
Rules {1, 2} have been declared as UpValues for Orth.
Rules {1, 2, 3, 4} have been declared as UpValues for Orth.
Rules {1, 2, 3, 4, 5, 6, 7, 8} have been declared as UpValues for Orth.
** MakeRule : Potential problems moving indices on the LHS.
Rules {1, 2, 3, 4, 5, 6, 7, 8} have been declared as UpValues for Met.
Rules {1} have been declared as UpValues for Met.
Rules {1, 2, 3, 4} have been declared as UpValues for avSMet.
```

```
In[10]:= PrintAs [SMet] ^= "h"
```

```
Out[10]= h
```

```
In[11]:= PrintAs [Orth] ^= "n"
```

```
Out[11]= n
```

Metric perturbation

Finally, we define the metric perturbations. These are given as follows.

```
In[12]:= DefMetricFields [Met,  $\delta$ Met, SMet]
```

Further ingredients

Tensor perturbations

In this calculation we want to assume only tensor perturbations. These are given by the following tensor field in xPand, where the first index describes the perturbation order:

```
In[13]:= EtSMet[LI[1], -α, -β]
```

```
Out[13]= (1) Eαβ
```

By the following definition, we set the perturbation of the conformally rescaled metric (which is the physical metric divided by the square of the scale factor) to the tensor perturbations only.

```
In[14]:= tensors = {δMet[LI[ord_], α_, β_] → 2 EtSMet[LI[ord], α, β]}
```

```
Out[14]= { δMetord_αβ → 2 Eordαβ }
```

Helper function

In the following, it will be helpful to define a function which performs a few steps automatically. This function first performs a conformal transformation from the conformally related, symmetric metric to the physical spacetime metric, then inserts the pure tensor perturbations, shows the result in a 3+1 decomposition and finally returns it.

```
In[76]:= pertexp[expr_] := Module[{term = expr}, term = Conformal[Met, MetaSMet2][term];
  term = SplitPerturbations[term, tensors, SMet];
  If[term != 0, Print[VisualizeTensor[term, SMet]]];
  Return[term]]
```

Perturbation of the Einstein tensor

Expected result

The tensor we need to calculate is the perturbation of the Einstein tensor with mixed indices. For convenience, we lower one index again after calculating the perturbation. The formula is actually well known:

```
In[22]:= Met[-α, -γ] × Perturbation [EinsteinCD [γ, -β], 1]
ptotal = ExpandPerturbation [%];
ttotal = pertexp[%]
```

```
Out[22]=  $\mathcal{G}_{\alpha\gamma} \Delta[G[\nabla]^\gamma_\beta]$ 
```

	Orth	SMet
Orth	0	0
SMet	0	${}^{(1)}E_{\alpha\beta}'' + 2 \left({}^{(1)}E_{\alpha\beta}' \right) \mathcal{H} + 2 \left({}^{(1)}E_{\alpha\beta} \right) k - D_\gamma D^\gamma {}^{(1)}E_{\alpha\beta}$

```
Out[24]=  ${}^{(1)}E_{\alpha\beta}'' + 2 \left( {}^{(1)}E_{\alpha\beta}' \right) \mathcal{H} + 2 \left( {}^{(1)}E_{\alpha\beta} \right) k - D_\gamma D^\gamma {}^{(1)}E_{\alpha\beta}$ 
```

Extracting the inverse metric

This can also be calculated in two steps, namely by perturbing the Einstein tensor with lower indices, and adding a correction term from the perturbation of the inverse metric, which is used to raise one index:

```
In[28]:= Perturbation [EinsteinCD [-α, -β], 1]
plower = ExpandPerturbation [%];
tlower = pertexp[%]
```

```
Out[28]=  $\Delta[G[\nabla]_{\alpha\beta}]$ 
```

	Orth	SMet
Orth	0	0
SMet	0	${}^{(1)}E_{\alpha\beta}'' + 2 \left({}^{(1)}E_{\alpha\beta}' \right) \mathcal{H} - 2 \left({}^{(1)}E_{\alpha\beta} \right) \mathcal{H}^2 - 4 \left({}^{(1)}E_{\alpha\beta} \right)' \mathcal{H} - D_\gamma D^\gamma {}^{(1)}E_{\alpha\beta}$

```
Out[30]=  ${}^{(1)}E_{\alpha\beta}'' + 2 \left( {}^{(1)}E_{\alpha\beta}' \right) \mathcal{H} - 2 \left( {}^{(1)}E_{\alpha\beta} \right) \mathcal{H}^2 - 4 \left( {}^{(1)}E_{\alpha\beta} \right)' \mathcal{H} - D_\gamma D^\gamma {}^{(1)}E_{\alpha\beta}$ 
```

```
In[35]:= -δMet[LI[1], -α, -ζ] × Met[ζ, γ] × EinsteinCD [-γ, -β]
pinvmet = ExpandPerturbation [%];
tinvmet = pertexp[%]
```

```
Out[35]=  $-G[\nabla]_{\gamma\beta} g^{\zeta\gamma} \delta\text{Met}^1_{\alpha\zeta}$ 
```

	Orth	SMet
Orth	0	0
SMet	0	$2 \left({}^{(1)}E_{\alpha\beta} \right) \mathcal{H}^2 + 4 \left({}^{(1)}E_{\alpha\beta} \right)' \mathcal{H} + 2 \left({}^{(1)}E_{\alpha\beta} \right) k$

```
Out[37]=  $2 \left( {}^{(1)}E_{\alpha\beta} \right) \mathcal{H}^2 + 4 \left( {}^{(1)}E_{\alpha\beta} \right)' \mathcal{H} + 2 \left( {}^{(1)}E_{\alpha\beta} \right) k$ 
```

We check that their sum indeed yields the perturbation we calculated above.

```
In[51]:= plower + pinvmet - ptotal ;
Expand[%];
ToCanonical [%];
Simplify [%];
ToCanonical [%]
```

```
Out[55]= 0
```

```
In[56]:= tlower + tinvmet - ttotal ;
Expand[%];
ToCanonical [%]
```

```
Out[58]= 0
```

Einstein tensor with lower indices

To gain a better insight in how this result has been calculated, we break it into single terms, which we then calculate one by one.

```
In[77]:= plower ;
Expand[%];
SeparateMetric [[][%];
ToCanonical [%]
pterm = List @@ %;
```

```
Out[80]= -\frac{1}{2} R[\nabla] \delta \text{Met}^1_{\alpha\beta} + \frac{1}{2} g_{\alpha\beta} g^{\nu\zeta} g^{\lambda\mu} R[\nabla]_{\nu\lambda} \delta \text{Met}^1_{\zeta\mu} - \frac{1}{2} g^{\nu\zeta} (\nabla_\alpha \nabla_\beta \delta \text{Met}^1_{\nu\zeta}) +
```

$$\frac{1}{2} g^{\nu\zeta} (\nabla_\zeta \nabla_\alpha \delta \text{Met}^1_{\beta\nu}) + \frac{1}{2} g^{\nu\zeta} (\nabla_\zeta \nabla_\beta \delta \text{Met}^1_{\alpha\nu}) - \frac{1}{2} g^{\nu\zeta} (\nabla_\zeta \nabla_\nu \delta \text{Met}^1_{\alpha\beta}) -$$

$$\frac{1}{2} g_{\alpha\beta} g^{\nu\zeta} g^{\lambda\mu} (\nabla_\mu \nabla_\zeta \delta \text{Met}^1_{\nu\lambda}) + \frac{1}{2} g_{\alpha\beta} g^{\nu\zeta} g^{\lambda\mu} (\nabla_\mu \nabla_\lambda \delta \text{Met}^1_{\nu\zeta})$$

Step-by-step calculation

Trace term

First note that by definition, the tensor perturbation is spatial and trace-free:

```
In[84]:= Met[\gamma, \zeta] \times Perturbation [Met[-\gamma, -\zeta], 1]
pertexp [%]
```

```
Out[84]= g^{\nu\zeta} \delta \text{Met}^1_{\nu\zeta}
```

```
Out[85]= 0
```

Hence, the same holds also for its second covariant derivative, and so the corresponding terms vanish.

```
In[86]:= Met[γ, ζ] × CD[-α][CD[-β][Perturbation [Met[-γ, -ζ], 1]]]
pertexp[%]
```

```
Out[86]= g^{γζ} (∇_α ∇_β δMet^1_{γζ})
```

```
Out[87]= 0
```

Contraction with Ricci tensor

For the cosmologically symmetric Friedmann-Lemaître-Robertson-Walker background, also the spatial components of the Ricci tensor are proportional to the background metric components.

```
In[135]:= RicciCD[-μ, -ν]
pertexp[%]
```

```
Out[135]= R[∇]_{μν}
```

	Orth	SMet
Orth	$-3 \dot{\mathcal{H}}$	0
SMet	0	$2 \mathcal{H}^2 h_{\mu\nu} + \dot{\mathcal{H}} h_{\mu\nu} + 2 h_{\mu\nu} k$

```
Out[136]= -3 \dot{\mathcal{H}} n_\mu n_\nu + 2 \mathcal{H}^2 h_{\mu\nu} + \dot{\mathcal{H}} h_{\mu\nu} + 2 h_{\mu\nu} k
```

Hence, their contraction with the tensor perturbation vanishes.

```
In[88]:= RicciCD[-μ, -ν] × Met[γ, μ] × Met[ζ, ν] × Perturbation [Met[-γ, -ζ], 1]
pertexp[%]
```

```
Out[88]= g^{γμ} g^{ζν} R[∇]_{μν} δMet^1_{γζ}
```

```
Out[89]= 0
```

Divergence term

Next, we use the fact that the tensor perturbation is also divergence-free. This allows us to calculate

```
In[90]:= Met[γ, ζ] × CD[-ζ][Perturbation [Met[-γ, -α], 1]]
pertexp[%]
```

```
Out[90]= g^{γζ} (∇_ζ δMet^1_{γ α})
```

```
Out[91]= 0
```

Reordered divergence terms

There are two more terms which appear like divergences, but their covariant derivatives are in the wrong order. Recall that changing the order of covariant derivatives introduces curvature tensors (and

in general also torsion tensors, but these vanish for the Levi-Civita connection):

```
In[96]:= CommuteCovDs [CD[-μ][CD[-α][Perturbation [Met[-ν, -β], 1]], CD, {-α, -μ}]
Expand[%]
```

$$\text{Out[96]}= -R[\nabla]_{\alpha\mu\nu}{}^{\gamma} \delta\text{Met}^1_{\gamma\beta} - R[\nabla]_{\alpha\mu\beta}{}^{\gamma} \delta\text{Met}^1_{\nu\gamma} + \nabla_{\alpha} \nabla_{\mu} \delta\text{Met}^1_{\nu\beta}$$

$$\text{Out[97]}= -R[\nabla]_{\alpha\mu\nu}{}^{\gamma} \delta\text{Met}^1_{\gamma\beta} - R[\nabla]_{\alpha\mu\beta}{}^{\gamma} \delta\text{Met}^1_{\nu\gamma} + \nabla_{\alpha} \nabla_{\mu} \delta\text{Met}^1_{\nu\beta}$$

Now the second order derivative term indeed reproduces the divergence which we previously calculated. For the remaining two terms, taking into account their coefficient in the perturbation of the Einstein tensor, we find the expressions

```
In[102]:= prediva1 = -Met[μ, ν] * RiemannCD[-α, -μ, -β, γ] * Perturbation [Met[-ν, -γ], 1] / 2
trediva1 = pertexp[%]
```

$$\text{Out[102]}= -\frac{1}{2} g^{\mu\nu} R[\nabla]_{\alpha\mu\beta}{}^{\gamma} \delta\text{Met}^1_{\nu\gamma}$$

	Orth	SMet
Orth	0	0
SMet	0	$\binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} k$

$$\text{Out[103]}= \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} k$$

```
In[104]:= prediva2 = -Met[μ, ν] * RiemannCD[-α, -μ, -ν, γ] * Perturbation [Met[-γ, -β], 1] / 2
trediva2 = pertexp[%]
```

$$\text{Out[104]}= -\frac{1}{2} g^{\mu\nu} R[\nabla]_{\alpha\mu\nu}{}^{\gamma} \delta\text{Met}^1_{\gamma\beta}$$

	Orth	SMet
Orth	0	0
SMet	0	$2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' + 2 \binom{(1)}{E_{\alpha\beta}} k$

$$\text{Out[105]}= 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' + 2 \binom{(1)}{E_{\alpha\beta}} k$$

Note that each of these terms appears once again, but with the opposite order of the free indices.

```
In[106]:= predivb1 = -Met[μ, ν] * RiemannCD[-β, -μ, -α, γ] * Perturbation [Met[-ν, -γ], 1] / 2
tredivb1 = pertexp[%]
```

$$\text{Out[106]}= -\frac{1}{2} g^{\mu\nu} R[\nabla]_{\beta\mu\alpha}{}^{\gamma} \delta\text{Met}^1_{\nu\gamma}$$

	Orth	SMet
Orth	0	0
SMet	0	$\binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} k$

$$\text{Out[107]}= \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} k$$


```
In[108]:= predivb2 = -Met[μ, ν] * RiemannCD[-β, -μ, -ν, γ] * Perturbation[Met[-γ, -α], 1]/2
tredivb2 = pertexp[%]
```

$$\text{Out[108]} = -\frac{1}{2} g^{\mu\nu} R[\nabla]_{\beta\mu\nu}{}^{\gamma} \delta\text{Met}^1_{\gamma\alpha}$$

	Orth	SMet
Orth	0	0
SMet	0	$2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' + 2 \binom{(1)}{E_{\alpha\beta}} k$

$$\text{Out[109]} = 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' + 2 \binom{(1)}{E_{\alpha\beta}} k$$

The result is the same, since the terms are symmetric in these indices. This is easily checked:

```
In[123]:= prediva1 - predivb1 ;
pertexp[%]
```

Out[124]= 0

```
In[125]:= prediva2 - predivb2 ;
pertexp[%]
```

Out[126]= 0

```
In[127]:= trediva1 - tredivb1
```

Out[127]= 0

```
In[128]:= trediva2 - tredivb2
```

Out[128]= 0

Ricci scalar term

For the term involving the Ricci scalar, we calculate first its background value.

```
In[131]:= RicciScalarCD []
pertexp[%]
```

Out[131]= R[∇]

$$\frac{6 \mathcal{H}^2}{a^2} + \frac{6 \mathcal{H}'}{a^2} + \frac{6 k}{a^2}$$

$$\text{Out[132]} = \frac{6 \mathcal{H}^2}{a^2} + \frac{6 \mathcal{H}'}{a^2} + \frac{6 k}{a^2}$$

Hence, its product with the metric perturbation, together with the coefficient in the Einstein tensor, is given by

```
In[133]:= pricsca = -RicciScalarCD [] * Perturbation [Met[-α, -β], 1] / 2
tricsca = pertexp [%]
```

$$\text{Out[133]} = -\frac{1}{2} R[\nabla] \delta \text{Met}^1_{\alpha\beta}$$

	Orth	SMet
Orth	0	0
SMet	0	$-6 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 - 6 \binom{(1)}{E_{\alpha\beta}} \dot{\mathcal{H}} - 6 \binom{(1)}{E_{\alpha\beta}} k$

$$\text{Out[134]} = -6 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 - 6 \binom{(1)}{E_{\alpha\beta}} \dot{\mathcal{H}} - 6 \binom{(1)}{E_{\alpha\beta}} k$$

Second derivative term

The last term in the perturbation of the lower index Einstein tensor involves second order covariant derivatives of the perturbations. Calculating these is more tedious, but straightforward.

```
In[139]:= psecond = -Met[γ, ζ] * CD[-γ][CD[-ζ][Perturbation [Met[-α, -β], 1]]] / 2
tsecond = pertexp [%]
```

$$\text{Out[139]} = -\frac{1}{2} g^{\gamma\zeta} (\nabla_\gamma \nabla_\zeta \delta \text{Met}^1_{\alpha\beta})$$

	Orth	SMet
Orth	0	0
SMet	0	$\binom{(1)}{E_{\alpha\beta}}'' + 2 \binom{(1)}{E_{\alpha\beta}}' \mathcal{H} - 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 - D_\gamma D^\gamma \binom{(1)}{E_{\alpha\beta}}$

$$\text{Out[140]} = \binom{(1)}{E_{\alpha\beta}}'' + 2 \binom{(1)}{E_{\alpha\beta}}' \mathcal{H} - 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 - D_\gamma D^\gamma \binom{(1)}{E_{\alpha\beta}}$$

Inverse metric term

For completeness, we recall once more the term obtained by raising one index of the Einstein tensor before calculating its perturbation. This is obtained from the background value of the Einstein tensor, which is given as follows.

```
In[141]:= EinsteinCD [-μ, -ν]
pertexp [%]
```

$$\text{Out[141]} = G[\nabla]_{\mu\nu}$$

	Orth	SMet
Orth	$3 \mathcal{H}^2 + 3 k$	0
SMet	0	$-\mathcal{H}^2 h_{\mu\nu} - 2 \dot{\mathcal{H}} h_{\mu\nu} - h_{\mu\nu} k$

$$\text{Out[142]} = 3 \mathcal{H}^2 n_\mu n_\nu - \mathcal{H}^2 h_{\mu\nu} - 2 \dot{\mathcal{H}} h_{\mu\nu} + 3 n_\mu n_\nu k - h_{\mu\nu} k$$

Indeed, here we recognize the term calculated earlier.

```
In[143]:= -δMet[LI[1], -α, -ζ] × Met[ζ, γ] × EinsteinCD[-γ, -β]
pertexp[%]
% - tinvmet
```

```
Out[143]:= - G[∇]γβ gζγ δMet1αζ
```

	Orth	SMet
Orth	0	0
SMet	0	$2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + 4 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' + 2 \binom{(1)}{E_{\alpha\beta}} k$

```
Out[144]:= 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 + 4 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' + 2 \binom{(1)}{E_{\alpha\beta}} k
```

```
Out[145]:= 0
```

Final result

Lower indices

We first check that we have correctly expanded the perturbation of the Einstein tensor with lower indices.

```
In[147]:= prediva1 + prediva2 + predivb1 + predivb2 + pricsca + psecond ;
pertexp[%]
% - tlower
```

	Orth	SMet
Orth	0	0
SMet	0	$\binom{(1)}{E_{\alpha\beta}}'' + 2 \binom{(1)}{E_{\alpha\beta}}' \mathcal{H} - 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 - 4 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' - D_\gamma D^\gamma \binom{(1)}{E_{\alpha\beta}}$

```
Out[148]:= \binom{(1)}{E_{\alpha\beta}}'' + 2 \binom{(1)}{E_{\alpha\beta}}' \mathcal{H} - 2 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}^2 - 4 \binom{(1)}{E_{\alpha\beta}} \mathcal{H}' - D_\gamma D^\gamma \binom{(1)}{E_{\alpha\beta}}
```

```
Out[149]:= 0
```

Mixed indices

Taking into account also the inverse metric term, we obtain the desired result.

```
In[153]:= prediva1 + prediva2 + predivb1 + predivb2 + pricsca + psecond + pinvmet ;
pertexp[%]
% - tttotal
```

	Orth	SMet
Orth	0	0
SMet	0	${}^{(1)}E_{\alpha\beta}'' + 2 \left({}^{(1)}E_{\alpha\beta}' \right) \mathcal{H} + 2 \left({}^{(1)}E_{\alpha\beta} \right) k - D_\gamma D^\gamma {}^{(1)}E_{\alpha\beta}$

Out[154]= ${}^{(1)}E_{\alpha\beta}'' + 2 \left({}^{(1)}E_{\alpha\beta}' \right) \mathcal{H} + 2 \left({}^{(1)}E_{\alpha\beta} \right) k - D_\gamma D^\gamma {}^{(1)}E_{\alpha\beta}$

Out[155]= 0